
Mplus code for Model 1: Multilevel bivariate VAR(1) model with random means and random lagged parameters

TITLE: Cogito data

DATA: file is CogitoOlder.dat;

VARIABLE:
NAMES       = id sessdate na1 na2 na3 na4 na5 na6 na7 na8 na9 na10
                pa1 pa2 pa3 pa4 pa5 pa6 pa7 pa8 pa9 pa10 sessionNr 
[bookmark: _GoBack]                age_pre sex CESDpre CESDpost dayNA dayPA older;
CLUSTER     = id;  
USEVAR      = dayPA dayNA;
LAGGED      = dayPA(1) dayNA(1);
TINTERVAL   = sessdate(1);
MISSING     = all(-999);

ANALYSIS:   TYPE IS TWOLEVEL RANDOM;
            ESTIMATOR = BAYES; 
            PROC = 2; BITER = (5000);
            BSEED = 5699; THIN = 10;

MODEL:
    %WITHIN%
    p_pp | dayPA ON dayPA&1; ! Random slope for PA_t regressed on PA_t-1 (autoregression)
    p_pn | dayPA ON dayNA&1; ! Random slope for PA_t regressed on NA_t-1 (cross-lagged) 
    p_np | dayNA ON dayPA&1; ! Random slope for NA_t regressed on PA_t-1 (cross-lagged)  
    p_nn | dayNA ON dayNA&1; ! Random slope for NA_t regressed on NA_t-1 (autoregression) 

    %BETWEEN%
    p_pp-p_nn dayPA dayNA WITH p_pn-p_nn dayPA dayNA; ! Correlations between all 6 random effects
    
OUTPUT: TECH1 TECH8 STDYX;

PLOT: TYPE = PLOT3; FACTOR =ALL;

Mplus code for Model 2: Multilevel bivariate VAR(1) model with random means, random lagged parameters, and random innovation variances and covariance

TITLE: Cogito data

DATA: file is CogitoOlder.dat;


VARIABLE:
NAMES       = id sessdate na1 na2 na3 na4 na5 na6 na7 na8 na9 na10
                pa1 pa2 pa3 pa4 pa5 pa6 pa7 pa8 pa9 pa10 sessionNr 
                age_pre sex CESDpre CESDpost dayNA dayPA older;
CLUSTER     = id;  
USEVAR      = dayPA dayNA;
LAGGED      = dayPA(1) dayNA(1);
TINTERVAL   = sessdate(1);
MISSING     = all(-999);

ANALYSIS:   TYPE IS TWOLEVEL RANDOM;
            ESTIMATOR = BAYES; 
            PROC = 2; BITER = (5000);
            BSEED = 5699; THIN = 10;

MODEL:
    %WITHIN%
    p_pp | dayPA ON dayPA&1; ! Random slope for PA_t regressed on PA_t-1 (autoregression)
    p_pn | dayPA ON dayNA&1; ! Random slope for PA_t regressed on NA_t-1 (cross-lagged) 
    p_np | dayNA ON dayPA&1; ! Random slope for NA_t regressed on PA_t-1 (cross-lagged)  
    p_nn | dayNA ON dayNA&1; ! Random slope for NA_t regressed on NA_t-1 (autoregression) 

    eta BY dayPA@1 dayNA@-1; ! Latent variable representing negative covariance between innovations of PA and NA
    logpsi | eta; ! Random negative covariance 
    logpi_PA | dayPA; ! Random unique innovation variance for PA
    logpi_NA | dayNA; ! Random unique innovation variance for NA

    %BETWEEN% 
    p_pp-p_nn logpi_PA logpi_NA logpsi dayPA dayNA WITH 
    p_pn-p_nn logpi_PA logpi_NA logpsi dayPA dayNA; ! Allow all 9 random effects to covary

OUTPUT: TECH1 TECH8 STDYX STAND(CLUSTER) FSCOMPARISON;

PLOT: TYPE = PLOT3; FACTOR =ALL;

Mplus code for Model 3: Multilevel bivariate VAR(1) model with random means, random lagged parameters, and random innovation variances and covariance, and between-person level predictor and outcome

TITLE: Cogito data

DATA: file is CogitoOlder.dat;

VARIABLE:
NAMES       = id sessdate na1 na2 na3 na4 na5 na6 na7 na8 na9 na10
                pa1 pa2 pa3 pa4 pa5 pa6 pa7 pa8 pa9 pa10 sessionNr 
                age_pre sex CESDpre CESDpost dayNA dayPA older;
CLUSTER     = id;  
USEVAR      = dayPA dayNA CESDpre CESDpost;
BETWEEN     = CESDpre CESDpost;
LAGGED      = dayPA(1) dayNA(1);
TINTERVAL   = sessdate(1);
MISSING     = all(-999);

DEFINE: CENTER CESDpre CESDpost (GRANDMEAN);

ANALYSIS:   TYPE IS TWOLEVEL RANDOM;
            ESTIMATOR = BAYES; 
            PROC = 2; BITER = (5000);
            BSEED = 5682; THIN = 10;

MODEL:
    %WITHIN%
    p_pp | dayPA ON dayPA&1; ! Random slope for PA_t regressed on PA_t-1 (autoregression)
    p_pn | dayPA ON dayNA&1; ! Random slope for PA_t regressed on NA_t-1 (cross-lagged) 
    p_np | dayNA ON dayPA&1; ! Random slope for NA_t regressed on PA_t-1 (cross-lagged)  
    p_nn | dayNA ON dayNA&1; ! Random slope for NA_t regressed on NA_t-1 (autoregression) 

eta BY dayPA@1 dayNA@-1; ! Latent variable representing negative covariance between innovations of PA and NA
    logpsi | eta; ! Random negative covariance 
    logpi_PA | dayPA; ! Random unique innovation variance for PA
    logpi_NA | dayNA; ! Random unique innovation variance for NA


    %BETWEEN%
    p_pp-p_nn dayPA dayNA ON CESDpre (a1-a6); ! Regress all 4 random slopes and 2 random means on between level predictor
    logpi_PA logpi_NA logpsi ON CESDpre (a7-a9); ! Regress all 3 random variances on between level predictor
    CESDpost ON p_pp-p_nn dayPA dayNA logpi_PA logpi_NA logpsi CESDpre (b1-b10); ! Regress between level outcome on all 9 random
   ! effects and the between level predictor 

 MODEL CONSTRAINT:
  new (ab_p_pp); ab_p_pp=a1*b1; ! Compute indirect effect through autoregression of PA
  new (ab_p_pn); ab_p_pn=a2*b2; ! Compute indirect effect through cross-lagged regression from NA to PA
  new (ab_p_np); ab_p_np=a3*b3; ! Compute indirect effect through cross-lagged regression from PA to NA
  new (ab_p_nn); ab_p_nn=a4*b4; ! Compute indirect effect through autoregression of NA
  new (ab_dayPA); ab_dayPA=a5*b5; ! Compute indirect effect through mean of PA
  new (ab_dayNA); ab_dayNA=a6*b6; ! Compute indirect effect through mean of NA
  new (ab_lvPA); ab_lvPA=a7*b7; ! Compute indirect effect through log of the unique innovation variance of PA
  new (ab_lvNA); ab_lvNA=a8*b8; ! Compute indirect effect through log of the unique innovation variance of NA
  new (ab_lCov); ab_lCov=a9*b9; ! Compute indirect effect through log of the negative innovation covariance of PA and NA

OUTPUT: TECH1 TECH8;

PLOT: TYPE = PLOT3; FACTOR =ALL;

